
SAS Codes for ARIMA, ARIMAX, ARCH and GARCH
N. Uttam Singh, Kishore K Das, Anjoo Yumnam and Aniruddha Roy, 2018

Code for scatter plot.

```
proc sgplot data=GROWTH_FORECAST;  
scatter y=Yld_Grth x=year;  
SERIES y=Yld_Grth x=year;  
run;
```

Code for identifying the correlation properties.

```
proc arima data=GROWTH_FORECAST ;  
identify var=Yld_Grth nlag=12;  
run;
```

Code for SCAN and ESACF method of order identification.

```
proc arima data=GROWTH_FORECAST;  
identify var=Yld_Grth scan esacf;  
run;
```

Code for choosing lag length of ADF test.

```
proc autoreg data=GROWTH_FORECAST;  
model Yld_Grth = YEAR / nlag=5 ;  
run;
```

Code for ADF unit root test.

```
proc arima data=GROWTH_FORECAST;  
identify var=Yld_Grth stationarity=(adf=5);  
run;
```

Code for various model estimation.

```
proc arima data=GROWTH_FORECAST;  
identify var=Yld_Grth (1);  
estimate p=2 q=2 ;  
run;
```

Code for forecast.

```
proc arima data=GROWTH_FORECAST;  
identify var=Yld_Grth ;  
estimate p=1 q=0 ;  
forecast lead=5  
out=results PRINTALL;  
run;
```

Code for ARIMAX

```

proc arima data=GROWTH_FORECAST;
identify var= neavg_tem ;
estimate q=1;
run;
identify var =YLD_Grth crosscorr=( neavg_tem );
run;
estimate input=( NErain );
run;
estimate p=1 q=1 input=( NErain )NOSTABLE maxiter=100;
run;
forecast lead=5 out=results printall;
run;
proc arima data=GROWTH_FORECAST;
identify var= neavg_tem ;
run;
identify var =YLD_Grth crosscorr=( neavg_tem );
run;
estimate input=( neavg_tem );
run;
estimate p=1 q=0 input=( neavg_tem )maxiter=100;
run;
forecast lead=5 out=results printall;
run;

```

Code for ARCH and GARCH

```

proc autoreg data=GROWTH_FORECAST ;
model YLD_Grth= / noint nlag=1 method=ml maxiter=100
archtest;
run;
quit;
proc autoreg data=GROWTH_FORECAST ;
model YLD_Grth= / nlag=1 archtest=(qlm);
model YLD_Grth= / nlag=1 archtest=(lk, wl);
run;
quit;

proc autoreg data=GROWTH_FORECAST ;
ar_1_noint : model YLD_Grth= / noint nlag=1 method=ml
maxiter=100 NOPRINT;
output out=result cev=vhat p=Predicted r=resi;
proc print data=result;
proc sgplot data=result;
TITLE "AR(1)-NO INTERCEPT: ACTUAL YIELD GROWTH VERSUS
PREDICTED";
SCATTER y=Yld_Grth x=year ;

```

```
SERIES y=Predicted x=year;  
run;  
quit;
```

```
proc autoreg data=GROWTH_FORECAST ;  
ar_1_garch_1_1_noint : model YLD_Grth= / noint nlag=1  
garch=(p=1,q=1) maxiter=100 NOPRINT;  
output out=result cev=vhat p=Predicted r=resi;  
proc print data=result;  
proc sgplot data=result;  
TITLE "AR(1)-GARCH(1,1)-NO INTERCEPT: ACTUAL YIELD GROWTH  
VERSUS PREDICTED";  
SCATTER y=Yld_Grth x=year ;  
SERIES y=Predicted x=year;  
run;  
quit;
```

```
proc autoreg data=GROWTH_FORECAST ;  
ar_2_garch_1_1_noint : model YLD_Grth= / noint nlag=2  
garch=(p=1,q=1) maxiter=100 ;  
output out=result cev=vhat p=Predicted r=resi;  
proc print data=result;  
proc sgplot data=result;  
TITLE "AR(2)-GARCH(1,1)-NO INTERCEPT: ACTUAL YIELD GROWTH  
VERSUS PREDICTED";  
SCATTER y=Yld_Grth x=year ;  
SERIES y=Predicted x=year;  
run;  
quit;
```